# $Id$
# Maintainer: Sven-Hendrik Haase <sh@lutzhaase.com>
# Contributor: Jameson Pugh <imntreal@gmail.com>

pkgname=sdl2_gfx
pkgver=1.0.3
pkgrel=1
epoch=1
pkgdesc="SDL graphics drawing primitives and other support functions (Version 2)"
url="http://www.ferzkopp.net/wordpress/2016/01/02/sdl_gfx-sdl2_gfx/"
arch=(i686 x86_64)
license=(zlib)
depends=(sdl2)
source=("http://www.ferzkopp.net/Software/SDL2_gfx/SDL2_gfx-${pkgver}.tar.gz")
sha512sums=('4efb48501d635492efcfe06b1d4691911ce58ffc34e3b622f2124ac6d8aa6994196ffb52f16ced4325db3ef6d93719516deab2bcf1ce934df9af06c2fecbdce5')

build() {
  cd SDL2_gfx-$pkgver
  ./configure --prefix=/usr
  make
}

package() {
  cd SDL2_gfx-$pkgver
  make DESTDIR="$pkgdir" install
  install -Dm644 COPYING "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}
